
7 ~ 8. Boolean Calculator
1. 소스

include Irvine32.inc

.data

Menu        byte    "1. x AND y", 13, 10        ; 메뉴를 표시할 문자열 

            byte    "2. x OR y", 13, 10 

            byte    "3. NOT x", 13, 10

            byte    "4. x XOR y", 13, 10

            byte    "5. Exit program", 13, 10

            byte    "Select ==> ", 0

equmsg      byte    " = ", 0                    ; 연산 결과를 표시할 때 사용되는 문자열

InputHexa   byte    "Input ==> ", 0             ; 입력 받을 때 사용되는 문자열

op1         DWORD   ?                           ; 첫번 째 입력 값

op2         DWORD   ?                           ; 두번 째 입력 값

cmd         DWORD   ?                           ; 메뉴 선택 입력 값

.code

main PROC                           ; main 함수

MenuLoop:

    call    PrintMenu               ; 메뉴 출력 함수

    sub     eax, 0

    jnz     MenuLoop                ; eax가 0이 아니면 계속 반복

    exit                            ; eax가 0이면 프로그램 종료

main ENDP

ReadHexa    PROC                    ; 화면에서 hexa 값을 읽는 함수

    mov     edx, offset InputHexa

    call    WriteString             ; 입력 메시지 출력

    call    ReadInt                 ; 사용자에게 숫자 하나 입력 받음

    ret

ReadHexa    ENDP

ReadHexa1   PROC                    ; 단항 연산자 입력 함수

    call    ReadHexa                ; 숫자 하나 입력 받음

    mov     op1, eax                ; op1에 저장

    ret

ReadHexa1   ENDP

ReadHexa2   PROC                    ; 이항 연산자 입력 함수

    call    ReadHexa                ; 숫자 하나 입력 받음

    mov     op1, eax                ; op1에 저장

    call    Readhexa                ; 숫자 하나 입력 받음

    mov     op2, eax                ; op2에 저장

    ret

ReadHexa2   ENDP

WriteOp     PROC                    ; 연산 결과 출력 함수

                                    ; 파라미터(출력할 정수 갯수, 정수1, 정수2,...)

    push    ebp                     ; ebp 스택에 저장

    mov     ebp, esp                ; ebp를 esp로 초기화

    mov     ecx, [ebp+08h]          ; ecx를 넘어온 정수 갯수 파라미터로 지정

    mov     esi, ebp

    add     esi, 0Ch                ; esi를 첫번째 정수 파라미터로 지정

L:

    mov     eax, [esi]

    call    WriteHex                ; 첫번째 정수를 16진수로 출력

    mov     edx, offset equmsg 

    call    WriteString             ; " = " 출력

    mov     eax, [esi]

    call    WriteBin                ; 첫번째 정수의 바이너리 출력

    call    Crlf                    ; 개행 문자 출력

    add     esi, 4                  ; 다음 숫자로 이동

    loop    L                       ; 루프 반복

    call    WaitMsg                 ; 대기 메시지 출력

    mov     esp, ebp                ; esp 복원

    pop     ebp                     ; ebp 복원

    ret

WriteOp     ENDP

AND_op      PROC                    ; AND 연산 함수

    call    ReadHexa2               ; 정수 두개 입력 받음

    mov     eax, op1

    and     eax, op2                ; eax = op1 & op2

    push    eax

    push    op2

    push    op1

    push    3

    call    WriteOp                 ; WriteOp(3, op1, op2, eax)

    add     esp, 10h                ; 스택 정리

    ret

AND_op      ENDP

OR_op       PROC                    ; OR 연산 함수

    call    ReadHexa2               ; 정수 두개 입력 받음

    mov     eax, op1

    or      eax, op2                ; eax = op1 | op2

    push    eax

    push    op2

    push    op1

    push    3

    call    WriteOp                 ; WriteOp(3, op1, op2, eax)

    add     esp, 10h                ; 스택 정리

    ret

OR_op       ENDP

NOT_op      PROC                    ; NOT 연산 함수

    call    ReadHexa1               ; 정수 하나 입력 받음

    mov     eax, op1

    not     eax                     ; eax = !op1

    push    eax

    push    op1

    push    2

    call    WriteOp                 ; WriteOp(2, op1, eax)

    add     esp, 0ch                ; 스택 정리

    ret

NOT_op      ENDP

XOR_op      PROC                    ; XOR 연산 함수

    call    ReadHexa2               ; 정수 두개 입력 받음

    mov     eax, op1

    xor     eax, op2                ; eax = op1 ^ op2

    push    eax

    push    op2

    push    op1

    push    3

    call    WriteOp                 ; WriteOp(3, op1, op2, eax)

    add     esp, 10h                ; 스택 정리

    ret

XOR_op      ENDP

PrintMenu PROC                      ; 메뉴 출력 함수

    call    ClrScr

    mov     edx, offset Menu

    call    WriteString             ; 메뉴 출력

    call    ReadInt                 ; 메뉴 입력 받음

    mov     cmd, eax                ; cmd에 저장

    cmp     eax, 1

    jb      Exit_PrintMenu          ; 입력한 숫자가 1보다 작으면 리턴

    cmp     eax, 5                      

    ja      Exit_PrintMenu          ; 입력한 숫자가 5보다 크면 리턴

    jmp     MenuTable[eax*4]        ; 1~5인 경우 적절한 곳으로 점프

_1:

    call    AND_op                  ; AND_op 수행

    jmp     Exit_PrintMenu          ; 리턴

_2:

    call    OR_op                   ; OR_op 수행

    jmp     Exit_PrintMenu          ; 리턴

_3:

    call    NOT_op                  ; NOT_op 수행

    jmp     Exit_PrintMenu          ; 리턴

_4:

    call    XOR_op                  ; XOR_op 수행

    jmp     Exit_PrintMenu          ; 리턴

_5:

    xor     eax, eax                ; eax = 0

    ret                             ; 리턴

Exit_PrintMenu:

    mov     eax, 1                  ; eax = 1

    ret                             ; 리턴

MenuTable   DWORD 0                 ; 메뉴 점프 테이블

            DWORD _1

            DWORD _2

            DWORD _3

            DWORD _4

            DWORD _5

PrintMenu ENDP

end main
2. 실행 결과

그림 1 프로그램 초기 실행 화면

그림 2 첫 번째 AND 메뉴 수행 화면

그림 3 두 번째 OR 메뉴 수행 화면

그림 4 세 번째 NOT 메뉴 수행 화면

그림 5 네 번째 XOR 메뉴 수행 화면
